Desenvolvemento en serie de funcións. Series numéricas

Aproximación dunha función mediante unha serie de potencias

O conxunto de monomios $\{x^n\}_{-\infty}^{\infty}$ constitúe unha base no espazo vectorial das funcións. Isto significa que toda función *f(x)* pódese aproximar en torno a un punto $x=a$ como unha serie de potencias (tamén chamada Polinomio de Taylor) de orde *n*:

$$f(x) \approx \sum_{i=1}^{n} a_i (x-a)^i, \quad x \in [a-\varepsilon, a+\varepsilon]$$

Isto é un polinomio en *(x-a)*. A aproximación terá un erro denotado por $O[(x-a)^{n+1}]$ que se le "erro de orde *n*". Este erro que será menor canto maior sexa *n* (porque o polinomio ten máis coeficientes e, polo tanto, máis graos de liberdade para aproximar a función *f*) e canto menor sexa $|x-a|$ (canto máis perto esteamos de $x=a$).

$$f(x) = \sum_{i=1}^{n} a_i (x-a)^i + O\left((x-a)^{n+1}\right) \quad x \in [a-\varepsilon, a+\varepsilon]$$

A serie de Taylor é unha serie particular, que toma a forma

$$\sum_{n=0}^{\infty} \frac{f^{(n)}(a)}{n!} (x-a)^n$$

onde $f^{(n)}(a)$ é á *n*-ésima derivada de *f* en $x=a$: é dicir, $a_n = f^{(n)}(a)/n!$.

Utilízanse as función **series(expr, x=a)** ou **series(expr, x=a, n)** para a aproximación dunha expresión nunha serie e a función **taylor(expr, x= a, n)** para a descomposición nunha **serie ou polinomio de Taylor**. Ambas realizan un desenvolvemento nunha serie de orde **n** (de Taylor no caso da función **taylor**) da expresión **expr** no punto **x=a**. Os argumentos son: **expr** a expresión a desenvolver, **x** o nome da variabel independente, **a** (x=0 se non se especifica o punto) o punto de descomposición e **n** a orde de descomposición.

```
> series(x/(1-x-x^2), x = 0);
```

$$x + x^2 + 2x^3 + 3x^4 + 5x^5 + O(x^6) \quad \textbf{(1.1.1)}$$

```
> series(x/(1-x-x^2), x); # se non se especifica o punto considérase x=0
```

$$x + x^2 + 2x^3 + 3x^4 + 5x^5 + O(x^6) \quad \textbf{(1.1.2)}$$

```
> s1:=series(x/(1-x-x^2), x, 10);
```

$$s1 := x + x^2 + 2x^3 + 3x^4 + 5x^5 + 8x^6 + 13x^7 + 21x^8 + 34x^9 + O(x^{10}) \quad \textbf{(1.1.3)}$$

```
> type(s1, 'series'); #comproba se s1 é de tipo serie
```

$$true \quad (1.1.4)$$

```
> type(s1, 'polynom'); # comproba se s1 é de tipo polinomio
```

$$false \quad (1.1.5)$$

```
> p1:=convert(s1, 'polynom'); # convirte a serie nun polinomio
```

$$p1 := x + x^2 + 2\,x^3 + 3\,x^4 + 5\,x^5 + 8\,x^6 + 13\,x^7 + 21\,x^8 + 34\,x^9 \quad (1.1.6)$$

```
> type(p1, 'polynom');
```

$$true \quad (1.1.7)$$

Esta serie ou polinomio de Taylor aproxima á función (neste caso $f(x)=x/(1-x-x^2)$) nun entorno de $x=0$: a aproximación é mellor cando máis perto esteamos de $x=0$ e canto meirande sexa a orde n do polinomio. Podemos comprobar esta aproximación co comando:

```
> f:=x/(1-x-x^2):plot([f,series(f,x=0,4)],x=-0.6..0.6);
```



Aumentando a orde da serie (no caso anterior, 4), acádase unha mellor aproximación. Nota que a diferencia aumenta cando nos alonxamos de $x=0$.

```
> series(x^3/(x^4+4*x-5), x = infinity);
```

$$\frac{1}{x} - \frac{4}{x^4} + \frac{5}{x^5} + \mathrm{O}\left(\frac{1}{x^7}\right) \quad (1.1.8)$$

```
> taylor(exp(x), x = 0, 6); #serie de Taylor
```

$$1 + x + \frac{1}{2}\,x^2 + \frac{1}{6}\,x^3 + \frac{1}{24}\,x^4 + \frac{1}{120}\,x^5 + \mathrm{O}(x^6) \quad (1.1.9)$$

```
> type(%, 'series');
```

$$true \quad (1.1.10)$$

```
> taylor(sin(x), x = Pi);
```

$$-(x-\pi) + \frac{1}{6}\,(x-\pi)^3 - \frac{1}{120}\,(x-\pi)^5 + \mathrm{O}\left((x-\pi)^6\right) \quad (1.1.11)$$

Podemos calcular o coeficiente i coa función **coeff(p, x^i)**:

```
> coeff(p1,x^2);
```

$$1 \quad (1.1.12)$$

onde $p1$ é o polinomio de Taylor.

Tamén se pode calcula-lo desenvolvemento en serie de Taylor de funcións de varias variábeis (p.ex. $f(x,y)$) con **mtaylor()**. Por exemplo, $f(x, y) =$ $e^{x^2+y^2}$ pódese desenvolver usando :

```
> mtaylor(exp(x^2+y^2),[x=0,y=0],8);
```

$$1+x^2+y^2+\frac{1}{2}x^4+y^2x^2+\frac{1}{2}y^4+\frac{1}{6}x^6+\frac{1}{2}y^2x^4+\frac{1}{2}y^4x^2+\frac{1}{6}y^6 \tag{1.1.13}$$

Suma de series numéricas

As series finitas numéricas e infinitas son da forma:

$$\sum_{k=1}^{n} a_k \qquad \sum_{n=1}^{\infty} a_n = \lim_{k\to\infty} \sum_{n=1}^{k} a_n$$

Utilízase o comando **sum** ou **Sum** (non evalúa a expresión) combinado con **value**.

sum(f, k= m..n) $\qquad \displaystyle\sum_{k=m}^{n} f$

onde:

f	-	expresión
k	-	nome do índice do sumatorio
m, n	-	extremos do sumatorio (enteiros ou expresións)

A función *sum( )* calcula a expresión simbólica dunha suma (finita ou infinita). Para calcular unha suma finita dun número constante de sumandos (non *k*, senón p.ex. 5 sumandos), mellor usa-lo comando *add(f, i = m..n)*. Exemplo:
$add(i^2, i=1..5)$

```
> Sum(1/k!, k=0..infinity); value(%);
```

$$\sum_{k=0}^{\infty} \frac{1}{k!}$$

$$e \tag{1.2.1}$$

```
> Sum(1/n^2,n=1..infinity)=sum(1/n^2,n=1..infinity);
```

$$\sum_{n=1}^{\infty} \frac{1}{n^2} = \frac{1}{6}\pi^2 \tag{1.2.2}$$

Productos de series

Os produtos finitos están definidos por

$$\prod_{i=1}^{n} a_i = a_1 a_2 \ldots a_n$$

E os produtos infinitos están definidos por

$$\prod_{n=1}^{\infty} a_n = \lim_{k \to \infty} \prod_{n=1}^{k} a_n$$

En Maple, estos productos calcúlanse coa función **product(f, k=1..n)**. A función **Product(f,k=1..n)** fai o mesmo pero só mostra a expresión. Por exemplo, para calcular:

$$\prod_{n=0}^{\infty} \left[1 - \frac{4}{(2\cdot n+1)^4}\right]$$

```
> Product(1-4/(2*n+1)^4,n=0..infinity)=product(1-4/(2*n+1)^4,n=0..
  infinity);
```

$$\prod_{n=0}^{\infty} \left(1 - \frac{4}{(2\,n+1)^4}\right) = \sin\left(\pi\left(\frac{1}{2} + \frac{1}{2}\,\mathrm{I}\sqrt{2}\right)\right)\sin\left(\pi\left(-\frac{1}{2}\sqrt{2} + \frac{1}{2}\right)\right) \qquad \textbf{(1.3.1)}$$